/*

Pia Dietze	
Proposal title: How Social Class and the Framing of Income Inequality Affect Solidarity Within & Across Groups


HYPOTHESES

Stated-Hyp1: How does the framing of economic inequality (as high-class privelege 
	vs. low-class disadvantage) affects individuals’ support 
	for redistributive policies and action to address inequality among both members 
	of the disadvantaged (lower-class) and priveleged (upper class) groups.

	Test-Hyp1: There will be a difference in the effect of framing inequality as 
	upper-class privelege vs. lower-class disadvantage on redistributive preferences. 	
	
	Test-Hyp2: There will be a difference in the effect of framing inequality as 
	upper-class privelege vs. lower-class disadvantage on liklihood of action to 
	address inequality. 	

	Test-Hyp3: In terms of redistributive preference, upper and lower
	class individuals will react differently to the two frames.	
	
	Test-Hyp4: In terms of liklihood of action to address inequality, upper and lower
	class individuals will react differently to the two frames.
	
Stated-Hyp2: The effect of the frame will be mediated by percieved injustice of 
	economic inequality and perceptions of similarily and solidarity between the 
	lower-classes and other minorities among lower-class respondents. Life hardship 
	beliefs and beliefs in personal privelege will be mediators for upper class 
	participants. (pages 5-6) 	
	
	Test-Hyp5: Within lower-class participants, the effect of the frame on redistributive preference will be mediated by percieved injustice of economic inequality (1 item).
 	
	Test-Hyp6: Within lower-class participants, the effect of the frame  on redistributive preference will be mediated by perceptions of similarity between lower-classes and other minorities (4-item scale).	
	
	Test-Hyp7: Within upper-class participants, the effect of the frame  on redistributive preference will be mediated by life hardship beliefs (3 items).
	
	Test-Hyp8: Within upper-class participants, the effect of the frame  on redistributive preference will be mediated by beliefs in personal privelege (2 items).
	
	Test-Hyp9: Within lower-class participants, the effect of the frame on the liklihood to act will be mediated by percieved injustice of economic inequality (1 item).
 	
	Test-Hyp10: Within lower-class participants, the effect of the frame on the liklihood to act will be mediated by perceptions of similarity between lower-classes and other minorities (4-item scale).	
	
	Test-Hyp11: Within upper-class participants, the effect of the frame on the liklihood to act will be mediated by life hardship beliefs (3 items).
	
	Test-Hyp12: Within upper-class participants, the effect of the frame on the liklihood to act will be mediated by beliefs in personal privelege (2 items).
	
	*/


clear all
use Dietze961.dta, clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATION

* framing as privelege of rich or disadvantage of poor
	clonevar frame = INFO
	lab def frame 1	"Lower-class disadvantage frame" 2	"Upper-class privilege frame"
	lab val frame frame
	tab frame, mis
	
* OUTCOME MEASURES

* support for redistributive policies
	// there is no measure of support for redistributive policies in these data.

	/*
q18
	Some people think that the government ought to reduce the income differences between the rich and the poor perhaps by raising the taxes of wealthy families or by giving income assistance to the poor. Others think the government should not concern itself with reducing this income difference between the rich and the poor. Think of a score of 1 meaning that the government should not concern itself with reducing income differences and a score of 7 as meaning that the government ought to reduce the income differences between rich and poor. What score between 1 and 7 comes closest to the way you feel?
Enter a score between 1 and 7: ______

q19
Some people think that major companies ought to reduce the pay differences between employees with high pay and those with low pay, perhaps by reducing the pay of executives or by increasing the pay of unskilled workers. Others think major companies should not concern themselves with reducing this pay difference between employees with high pay and those with low pay. Think of a score of 1 meaning that major companies should not concern themselves with reducing pay differences and a score of 7 as meaning that major companies ought to reduce the pay differences between employees with high pay and those with low pay. What score between 1 and 7 comes closest to the way you feel? 
Enter a score between 1 and 7: ______

*/

	replace Q18=. if Q18>7
	replace Q19=. if Q19>7
	gen redist_support= Q18+Q19
	tab redist_support, mis
	
* action to address inequality

	* sharing information
	clonevar act_shareinfo= Q16A 
	replace act_shareinfo=. if Q16A >7 // missing values
	tab act_shareinfo, missing
	
	* liklihood to protest
	clonevar act_protest= Q16B
	replace act_protest=. if Q16B >7 // missing values
	tab act_protest, missing	
	
	* combining into index 
	gen action = act_shareinfo + act_protest
	tab action, mis
	
* MODERATOR
	* respondent's subjective social class
	tab Q1 // no middle class
	recode Q1 (1/2=0) (4/5=1), gen(upperclass)
	lab def class 0 "lower class" 1 "upper class"
	lab val upperclass class
	tab upperclass, mis
	
* MEDIATORS

* percieved injustice (1 item)
	// please see Quex for details
	tab Q3 // higher values indicate stronger perception of injustice
	recode Q3 (7=1) (6=2) (5=3) (4=4) (3=5) (2=6) (1=7) (*=.), gen(percieved_unjust)
	lab def ineq 1 "strongly disagree" 7 "strongly agree"
	lab val percieved_unjust ineq
	tab percieved_unjust, mis
	
* percieved similarity with lower-class and minorities (4-item)
	// please see Quex for details
	foreach var of varlist Q4* {
		tab `var'
	recode `var' (7=1) (6=2) (5=3) (4=4) (3=5) (2=6) (1=7) (*=.)
	lab val `var' ineq
	tab `var', mis
	}	
	gen percieved_similarity= Q4A + Q4B + Q4C + Q4D
	tab percieved_similarity, mis


* life hardship beliefs (3 items)
	// please see Quex for details
	foreach var of varlist Q11* {
		tab `var', mis
	recode `var' (7=1) (6=2) (5=3) (4=4) (3=5) (2=6) (1=7) (*=.)
	lab val `var' ineq
	tab `var', mis
	}	
	
	/*NOTE: 
	Q11C is correctly-coded: higher values indicate R thinks their life has not been easy
	[My life has been easy.] Please indicate how much you agree or disagree with the following statements.
	- we need to re-reverse this item
	*/
	recode Q11C (7=1) (6=2) (5=3) (4=4) (3=5) (2=6) (1=7) (*=.)
	
	gen percieved_hardship = Q11A + Q11B + Q11C
	tab percieved_hardship, mis
	
	
* beliefs in personal privelege (2 items) 
	// please see Quex for details
	gen percieved_privelege= Q11D + Q11E
	tab percieved_privelege, mis

	
********************************************************************************

* Analysis

*	Test-Hyp1: There will be a difference in the effect of framing inequality as upper-class privilege vs. lower-class disadvantage on redistributive preferences. 	
	reg redist_support i.frame
	
	tess 2.frame, init(Dietze961) bonf(2)
	
	// reject. p=0.029  -Bonferroni corrected for 2 tests
	
*	Test-Hyp2: There will be a difference in the effect of framing inequality as upper-class privilege vs. lower-class disadvantage on liklihood of action to address inequality. 	
	reg action i.frame
	tess 2.frame, bonf(2)
	// reject. p=0.032.  -Bonferroni corrected for 2 tests
	
*	Test-Hyp3: In terms of redistributive preference, upper and lower class individuals will react differently to the two frames.	
	reg redist_support i.frame##i.upperclass
	tess 2.frame#1.upperclass, bonf(2)	
	// reject. 0.749
	
*	Test-Hyp4: In terms of liklihood of action to address inequality, upper and lower class individuals will react differently to the two frames.
	reg action i.frame##i.upperclass
	tess 2.frame#1.upperclass, bonf(2)	

	// reject. .930 // note (from JF): this was missing in original do file
	
*	Test-Hyp5: Within lower-class participants, the effect of the frame on redistributive preference will be mediated by percieved injustice of economic inequality (1 item).
 	
	reg redist_support frame if upperclass==0
	est sto m1
	reg redist_support frame percieved_unjust if upperclass==0	
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject. 0.144 	
	
	
*	Test-Hyp6: Within lower-class participants, the effect of the frame  on redistributive preference will be mediated by perceptions of similarity between lower-classes and other minorities (4-item scale).	
	
	est sto clear
	reg redist_support frame if upperclass==0
	est sto m1
	reg redist_support frame percieved_similarity if upperclass==0	
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject.  0.9764		
	
*	Test-Hyp7: Within upper-class participants, the effect of the frame  on redistributive preference will be mediated by life hardship beliefs (3 items).

	est sto clear
	reg redist_support frame if upperclass==1
	est sto m1
	reg redist_support frame percieved_hardship if upperclass==1
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject:  0.685
	
*	Test-Hyp8: Within upper-class participants, the effect of the frame  on redistributive preference will be mediated by beliefs in personal privelege (2 items).
		
	est sto clear
	reg redist_support frame if upperclass==1
	est sto m1
	reg redist_support frame percieved_privelege if upperclass==1
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject. 	0.6324 // note: the p-value looks to be .898
	
*	Test-Hyp9: Within lower-class participants, the effect of the frame on the liklihood to act will be mediated by percieved injustice of economic inequality (1 item).
 
	reg action frame if upperclass==0
	est sto m1
	reg action frame percieved_unjust if upperclass==0	
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject. 0.2207 
 
*	Test-Hyp10: Within lower-class participants, the effect of the frame on the liklihood to act will be mediated by perceptions of similarity between lower-classes and other minorities (4-item scale).	

	est sto clear
	reg action frame if upperclass==0
	est sto m1
	reg action frame percieved_similarity if upperclass==0	
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject.  0.8441	
	
*	Test-Hyp11: Within upper-class participants, the effect of the frame on the liklihood to act will be mediated by life hardship beliefs (3 items).

	est sto clear
	reg action frame if upperclass==1
	est sto m1
	reg action frame percieved_hardship if upperclass==1
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject:  0.7487	
	
*	Test-Hyp12: Within upper-class participants, the effect of the frame on the liklihood to act will be mediated by beliefs in personal privelege (2 items).	
	
	est sto clear
	reg action frame if upperclass==1
	est sto m1
	reg action frame percieved_privelege if upperclass==1
	est sto m2
	suest m1 m2
	test [m1_mean]frame=[m2_mean]frame

	tess, mediation
	
	// reject. 0.6058	
	
	
